# $Id$
# Maintainer: Andreas Radke <andyrtr@archlinux.org>
# Contributor: rabyte <rabyte.at.gmail.dot.com>
# Contributor: Johannes Krampf <wuischke.at.amule.dot.org>

pkgname=midori
pkgver=0.3.6
pkgrel=1
pkgdesc="A lightweight web browser based on Gtk WebKit"
arch=('ppc')
url="http://www.twotoasts.de/index.php?/pages/midori_summary.html"
license=('LGPL2.1')
install=midori.install
depends=('libwebkit>=1.3.13' 'libunique' 'hicolor-icon-theme' 
				'desktop-file-utils' 'libnotify' 'libxss' 'sqlite3' 'glib2'
				'gtk2')
makedepends=('libxml2' 'intltool' 'librsvg' 
						'python2>=2.7' 'vala' 'pkg-config')
optdepends=('vala: for use in extensions')
source=(http://archive.xfce.org/src/apps/$pkgname/0.3/$pkgname-$pkgver.tar.bz2)
md5sums=('c4cb0686601b1c470c317de3d3f8e8fd')

build() {
  cd ${srcdir}/${pkgname}-${pkgver}
  # python2 
  export PYTHON=/usr/bin/python2
  sed -i -e "s|#![ ]*/usr/bin/env python$|#!/usr/bin/env python2|" wscript waf
  
  ./configure --prefix=/usr \
	--enable-libnotify \
	--enable-addons \
	--disable-hildon \
	--enable-unique \
	--enable-nls \
	--enable-docs \
	--disable-apidocs
  make
}

package() {
  cd ${srcdir}/${pkgname}-${pkgver}
  make DESTDIR=${pkgdir} install
}
